
11. MORE ON FUNCTION PARAMETERS   

There are some very powerful lisp constructs for parameters to functions. They allow declaration of some 
parameters as optional, definition of default values and indefinite number of parameters.  

11.1. Optional Parameters 

In some situations, we want to make some parameters optional and give them a default value. This allows the 
user to pass a smaller number of parameters while still giving them the option of modifying the behaviour of the 
function using optional parameters.   

(defun foo (a b &optional c d) 
  (list a b c d )) 
> (foo 1 2) ⇒ (1 2 nil nil) 
> (foo 1 2 3 4) ⇒ (1 2 3 4) 

The values of c and d here are optional. They take the value of nil if no value is supplied  

You can also supply default values to optional parameters.  

(defun bar (a b &optional (c 10) (d 12)) 
  (list a b c d)) 
> (bar 1 2) ⇒ (1 2 10 12) 
> (bar 1 2 3 4) ⇒ (1 2 3 4) 

To see if an optional value was actually passed use this syntax:  

(defun mag (a &optional (b a b-supplied-p)) 
  (list a b b-supplied-p)) 
> (mag 1 2) ⇒ (1 2 T) 
> (mag 1) ⇒ (1 1 nil)  

The default value of b is taken to be the same as a and the b-supplied-p variable is set to T if a value for b 
was passed to the procedure. If it was not, the value for b-supplied-p is set to nil.  

11.2. Named Parameters or Keys 

Sometimes, it is useful to pass parameters based on names instead of based on their order. Keys provide a way to 
do this. For example: 

(defun foo1 (&key x y z)  
  (list x y z)) 

This defines a function that takes three parameters. The correct way to use it is  

> (foo1 :x 1 :y 2 :z 5) ⇒ (1 2 5) 

The following call will result in an error:  

> (foo1 1 2 3) 

You can also name the variables differently from the keys using this syntax 

(defun foo2 (&key   
             ((:a x) 10) ;; the 10 is the default value for x  
             ((:b y)))   ;; actually, it's a (( )) instead of a ( )  
                         ;; because a default value is also possible  
  (list x y))  
> (foo2 :a 34 :b 24) ⇒ (34 24) 
> (foo2 :b 12) ⇒ (10 12) 

As you can see, using keys automatically makes the parameters optional and you can provide a default value for 
a parameter using the key construct. 

11.3. &rest of the parameters  

Sometimes, you want to write a function with an indefinite number of arguments. An obvious example is the + 
function which can work on any number of arguments.  

One way to do that is to define a large number of optional parameter s in the function argument list. That, 
however, is very inefficient and tedious.  

The answer is &rest keyword. When you use the &rest keyword, all the parameters following that position 
are combined in a list which becomes the value of the &rest variable.  

(defun square-all (&rest numbers) 
  (mapcar #'square numbers)) 
 
> (square-all 1 2 3) ⇒  (1 4 9) 

1 2 and 3 are put in a list. mapcar applies the square function on the elements and returns a list as its value. 


